16
1

Appendix D TC  "Appendix D" \l 1 
Power plant

D 1. Type 45 Power plant performance 
D 1.1 Brief description of the TRNSYS component 
The Type 45 simulates the performance of a steam cycle power plant. The calculation is based on a number of empirical relations that fit the actual behavior of a power plant, therefore the calculation is not generally applicable, but adapted to the performance of the Columbia Generating Station. 
The values that were used for the calculation were obtained from a heat rate test. The curve fit parameters used are:

Standard heat rate:
AH=282.2 MW-Btu/kWh  
BH=6.282 Btu/kWh  





CH=0.002335 Btu/kWh-MW

Steam flow rate:
AL=-206.15 klb/hr  

BL=7.058 klb/hr-MW

Base auxiliary power:
AA=9.05 MW  

BA=0.0278 MW/MW

Boiler efficiency:
ABO=0.8703  

BBO=-0.00004525 1/MW

It has to be noted that the routine is suited only for calculation of power plant performance for a power plant with a maximum load of 535 MW and a maximum steam flow rate of 3,800 klb/hr. 

Following is the list of parameters, inputs and outputs that need to be specified:

Parameters:

1
First coefficient for standard heat rate 

[MW-Btu/kWh]

2
Second coefficient for standard heat rate

[Btu/kWh]

3
Third coefficient for standard heat rate

[Btu/kWh-MW]

4
Intercept for steam flow rate relation

[klb/hr]

5
Linear coefficient for steam flow rate relation
[klb/hr-MW]

6
Intercept for base auxiliary power

[MW]

7
Linear coefficient for base auxiliary power
[MW/MW]

8
Intercept for boiler efficiency


[]

9
Linear coefficient for boiler efficiency

[1/MW]

6
Fuel cost





[$/MMBtu]

Inputs:

1
Back pressure




["Hg]

2
Unit Load





[MW]

3
Auxiliary power




[MW]

4
Corresponding load at standard condition
[MW]

Outputs:

1
Overall heat rate




[Btu/kWh]

2
Corresponding heat rate at standard condition
[Btu/kWh]

3
Absolute change in net heat rate due to back
[Btu/kWh]


pressure

5
Absolute change in net heat rate due to 

[Btu/kWh]


auxiliary power for cooling

6
Fuel cost per hour



[$/hr]

7
Fuel cost per energy output


[$/MWh]

8
Corresponding load at standard condition
[MW]

For iteration purposes, input 4 has to be set equal to output 8 when the component is used in a TRNSYS deck. 

D 1.2 Source code of TYPE 45
      SUBROUTINE TYPE45(TIME,XIN,OUT,T,DTDT,PAR,INFO,ICNTRL,*) 

**************************************************************************

*  THIS TYPE SIMULATES THE PERFORMANCE OF A POWER PLANT. THE PERFORMANCE

*  IS CALCULATED FROM CORRECTION CURVES GIVEN IN THE ASME PTC 6.1 TURBINE

*  TEST CODE. INPUTS ARE THE  UNIT LOAD [MW], AUXILIARY POWER [MW] 

*  BACK PRESSURE ["Hg] AND THE EUIVALENT LOAD AT STANDARD CONDITIONS [MW].

*  OUTPUTS ARE:

*  1 THE OVERALL NET HEATRATE [BTU/kWh]

*  2 THE NET HEAT RATE IF THE BACK PRESSURE WERE 1.5"Hg RATE [BTU/kWh]

*  3 THE ABSOLUTE CHANGE IN NET HEAT RATE DUE TO BACK PRESSURE [BTU/kWh]   

*  4 THE ABSOLUTE CHANGE IN NET HEAT RATE DUE COOLING AUXILIARY POWER 

*  










[BTU/kWh]

*  5 THE OPERATING COST PER HOUR [$/hr]

*  6 THE COST PER ENERGY OUTPUT [$/MWh], 

*  7 THE CORRESPONDING STANDARD LOAD [MW].  

*  THE COSTS TAKE ONLY THE FUEL COSTS INTO ACCOUNT.

*  PARAMETERS ARE CURVE FIT COEFFICIENTS FOR STANDARD HEAT RATE, STEAM 

*  FLOW AND FUEL COSTS IN $/MBTU.

************************************************************************** 

**************************************************************************  

*  NOMENCLATURE

*  AH,BH,CH:  
CURVE FIT COEFFICIENTS FOR STANDARD HEAT RATE COMPUTATION, 

* 


BASED ON STANDARD GROSS LOAD

*  AL,BL:  
CURVE FIT COEFFICIENTS FOR STEAM FLOW RATE CALCULATION FROM 

*  


LOAD,BASED ON STANDARD GROSS LOAD

*  AA,BA:  
CURVE FIT COEFFICIENTS FOR BASE AUXILIARY POWER, BASED ON 

*  


GROSS LOAD

*  ABO,BBO:
CURVE FIT COEFFICIENTS FOR BOILER EFFICIENCY, BASED ON GROSS 

*


LOAD

*  A1,A2,A3,A4,A5:  CURVE FIT COEFFICIENTS FOR BACK PRESSURE




CORRECTION OF HEAT RATE

*  EFF_BO: 
BOILER EFFICIENCY 

*  GSTH:   
GROSS STANDARD TURBINE HEAT RATE, IF THE BACK PRESSURE WERE 

* 


1.5"Hg 






[BTU/kWh]

*  GTHR:   
ACTUAL GROSS TURBINE HEAT RATE  [BTU/kWh] 

*  GHR:    
GROSS PLANT HEAT RATE [BTU/kWh]

*  NHR_C:  
NET PLANT HEAT RATE FROM BASE AUXILIARY POWER [BTU/kWh]

*  NHR:    
NET PLANT HEAT RATE INCLUDING COOLING AUXILIARY POWER 

*  










[BTU/kWh]

*  ADHR_C: 
ABSOLUTE INCREMENT IN HEAT RATE DUE TO COOLING POWER 

*


CONSUMPTION 





[BTU/kWh]

*  NHR_S:  
NET PLANT HEAT RATE IF THE BACK PRESSURE WERE 1.5"Hg [BTU/kWh]

*  ADHR_S: 
ABSOLUTE INCREMENT IN HEAT RATE DUE TO BACK PRESSURE [BTU/kWh]

*  F:      
FUEL COST [$/kWh]

*  PB:     
BACK PRESSURE ["Hg]

*  MST:    
THROTTLE STEAM FLOW RATE [MMlb/hr]

*  NLOAD:  
NET PLANT LOAD [MW]

*  GLOAD:  
GROSS TURBINE LOAD  [MW]

*  LOAD_ST:
EQUIVALENT LOAD AT STANDARD CONDITIONS, INPUT  [MW]

*  LOAD_NE:
EQUIVALENT LOAD AT STANDARD CONDITIONS, OUTPUT [MW]

*  AUX:    
BASE AUXILIARY POWER [MW]

*  AUX_OLD:
BASE AUXILIARY POWER, ITERATION VARIABLE  [MW]  

*  AUX_C:  
AUXILIARY POWER DUE TO COOLING DEVICES [MW]

*  OP:     
OPERATING COST PER MWh OUTPUT  [$/MWh]

*  OPHR:   
TOTAL OPERATING COST PER HOUR  [$/hr] 

*  XIN:    
INPUT ARRAY

*  OUT:    
OUTPUT ARRAY

*  PAR:    
PARAMETER ARRAY

*  TIME,T,DTDT: TRNSYS SPECIFIC, NOT USED IN THIS CALCULATION 

****************************************************************************

      INTEGER INFO,ICNTRL       !TRNSYS specific                                    

      DOUBLE PRECISION XIN,OUT  !TRNSYS specific

      REAL TIME,T,DTDT,PAR      !TRNSYS specific 

      DOUBLE PRECISION A1,A2,A3,A4,A5,AH,BH,CH,AL,BL,AA,BA,ABO,BBO,

     .                 AUX,AUX_C,OP,OPHR,F,PB,MST,NLOAD,GLOAD,

     .                 LOAD_ST,LOAD_NE,DHR_PB,EFF_BO,GSTH,GTHR,

     .                 GHR,NHR,NHR_C,ADHR_C,NHR_S,ADHR_S

      DIMENSION XIN(4),OUT(7),PAR(10),INFO(15) !TRNSYS specific

C  PARAMETERS     

      AH=PAR(1)  !Btu-MW/kWh      

      BH=PAR(2)  !Btu/kWH

      CH=PAR(3)  !Btu/kWh-MW 

      AL=PAR(4)  !MMlb/hr

      BL=PAR(5)  !MMlb/hr-MW   

      AA=PAR(6)  !MW

      BA=PAR(7)  !MW/MW   

      ABO=PAR(8) !

      BBO=PAR(9) !1/MW

      F=PAR(10)  !$/MMBTU

C  INPUT VARIABLES             

      PB=XIN(1)

      NLOAD=XIN(2)

      AUX_C=XIN(3)

C  NEEDED FOR ITERATION

      LOAD_ST=XIN(4)

C  CALCULATE STEAM FLOW FROM STANDARD LOAD

      MST=(AL+BL*LOAD_ST)/1000 !Mlb/HR

C  CURVE FIT FOR HEAT RATE CORRECTION FROM ASME PTC 6.1 

      A1=36.282

      A2=1.058

      A3=6.636

      A4=-0.184

      A5=-35.897

      DHR_PB=A1*(PB-1.5)+A2*(PB-1.5)**2.+A3*(PB-1.5)*MST+

     .       A4*(PB-1.5)**2.*MST+A5*(PB-1.5)*MST**0.4

C  GROSS LOAD FROM NET LOAD, BASE AND COOLING AUXILIARY POWER

      GLOAD=(NLOAD+AA+AUX_C)/(1-BA)                             

C  AUXILIARY POWER BASED ON GROSS LOAD, IMPLICIT IN STATEMENT ABOVE         

      AUX=AA+BA*GLOAD 

C  BOILER EFFICIENCY BASED ON GROSS LOAD

      EFF_BO=ABO+BBO*GLOAD      

C  GROSS STANDARD TURBINE HEAT RATE AT 1.5"Hg BACK PRESSURE

      GSTH=(AH/LOAD_ST+BH+CH*LOAD_ST)*1000. 

C  ACTUAL GROSS TURBINE HEAT RATE FROM HEAT RATE CORRECTION

      GTHR=GSTH*(1.+DHR_PB/100.)                                             

C  GROSS PLANT HEAT RATE FROM BOILER EFFICIENCY     

      GHR=GTHR/EFF_BO

C  NET PLANT HEAT RATE, WITHOUT AUXILIARY POWER FOR COOLING

      NHR_C=GHR*GLOAD/(GLOAD-AUX)

C  NET PLANT HEAT RATE, INCLUDING BASE AND COOLING AUXILIARY POWER

      NHR=GHR*GLOAD/(GLOAD-AUX-AUX_C)         

C  ABSOLUTE INCREMENT IN HEAT RATE DUE TO AUXILIARY COOLING

      ADHR_C=NHR-NHR_C  

C  NET HEAT RATE IF THE BACK PRESSURE WERE 1.5"Hg

      NHR_S=NHR/(1.+DHR_PB/100.)

C  ABSOLUTE INCREMENT IN HEAT RATE DUE TO BACK PRESSURE

      ADHR_S=NHR-NHR_S           

C  NEW EQUIVALENT LOAD AT STANDARD CONDITION, INPUT FOR CONDENSER UNIT

C  FOR ITERATION

      LOAD_NE=GLOAD/(1.-DHR_PB/100.)       

C  OPERATING COST OF PLANT IN $/hr, ASSUMING CONSTANT

C  PLANT LOAD OVER ONE HOUR

      OPHR=NLOAD*NHR/1000.*F !$/hr  

C  COST IN $/MWh OUTPUT

      OP=NHR*F/1000. !$/MWh  

C  OUTPUT VARIABLES

      OUT(1)=NHR      ![Btu/kWh]

      OUT(2)=NHR_S    ![Btu/kWh]

      OUT(3)=ADHR_S   ![Btu/kWh]

      OUT(4)=ADHR_C   ![Btu/kWh]

      OUT(5)=OPHR     ![$/hr]

      OUT(6)=OP       ![$/MWh] 

      OUT(7)=LOAD_NE  ![MW]

      RETURN 1     

      END

D 2 Type 55 Control unit

D 2.1 Description of Type 55 control unit
The Type 55 is an auxiliary component that is used for the simulation of the Columbia Generating station. It is not generally applicable but is specifically written for this simulation. The component is used to generate volumetric flow rates for example for the pond mode, mixed plant inlet and discharge temperatures for cooling tower, cooling pond and power plant and the auxiliary power consumption of the cooling cycle. 

Following is the description of parameters, inputs and outputs

Parameters:
1
Volumetric flow rate for one condenser water 
[ft3/hr]


pump, unit 1

2
Flow rate ratio of one to two pump operation,


unit 1

3
Volumetric flow rate for one condenser water 
[ft3/hr]


pump, unit 2

4
Flow rate ratio of one to two pump operation,


unit 2

5
Tower operation parameter

6
Volumetric flow rate for one tower pump, 
[ft3/hr]


tower A

7
Flow rate ratio of one to two pump operation,


tower A

8
Volumetric flow rate for one tower pump, 
[ft3/hr]


tower B

9
Flow rate ratio of one to two pump operation,


tower B

10
Number of tower cells, tower A

11
Number of tower cells, tower B

12
Average fan power for tower A

13
Average fan power for tower B

14
Pump power fro one tower pump 

[MW]

15
Pump power for one make up pump

[MW]

16
Volumetric flow rate per make up pump

[ft3/hr]

17
Volumetric flow rate per low head pump
[ft3/hr]

Inputs:

1
Difference in temperature rise between low 
[˚F]


head cycle and main cooling cycle

2
Number of pumps for condenser 1

3
Number of pumps for condenser 2

4
Number of make up pumps

5
Number of low head pumps

6
Control variable from automatic controller

7
Tower A outlet temperature


[˚F]

8
Tower B outlet temperature


[˚F]

9
Pond outlet temperature



[˚F]

10
Condenser outlet temperature, unit 1

[˚F]

11
Condenser outlet temperature, unit 2

[˚F]

12
Auxiliary power, dummy 1


[MW]

13
Auxiliary power, dummy 1


[MW]

Outputs:

1
Mass flow rate to tower A


[kg/hr]

2
Mass flow rate to tower B


[kg/hr]

3
Volumetric flow rate to lake


[ft3/hr]

4
Make up flow rate



[ft3/hr]

5
Flow rate to condenser 1



[ft3/hr]

6
Flow rate to condenser 2



[ft3/hr]

7
Mixed plant discharge temperature

[˚F]

8
Mixed tower outlet temperature


[˚F]

9
Mixed plant inlet temperature


[˚F]

10
Auxiliary power consumption


[MW]

11
Number of tower pumps

D 2.2 Source code Type 55 Control unit

      SUBROUTINE TYPE55(TIME,XIN,OUT,T,DTDT,PAR,INFO,ICNTRL,*)

**************************************************************************

*  This Type serves as the control type for the cooling cycle simulation.

*  Parameters and inputs specify the plant operation mode, while the 

*  program converts this information into the desired input format

*  for the cooling cycle elements, which are cooling towers, condensers,

*  turbines, cooling pond and two dummy units.

**************************************************************************      

**************************************************************************

*  NOMENCLATURE

*  DENS: 
DENSITY OF WATER, ASSUMED TO BE CONSTANT AT 62.2 lb/ft^3

*  VWCD1: 
DESIGN COOLING WATER FLOW TO CONDENSER 1  [ft^3/hr]

*  VWCP1: 
COOLIN WATER FLOW TO CONDENSER 1 AT 1 PUMP  [ft^3/hr]

*  VWCD2: 
DESIGN COOLING WATER FLOW TO CONDENSER 2  [ft^3/hr]

*  VWCP2: 
COOLIN WATER FLOW TO CONDENSER 2 AT 1 PUMP  [ft^3/hr]

*  RAC1: 
RATIO OF ACTUAL TO DESIGN FLOW RATE FOR 1 PUMP OPERATION,

*           
CONDENSER 1. DESIGN MEANS TWO TOWER OPERATION

*  RAC2: 
SAME AS RAC1, FOR CONDENSER 2

*  VWCO1: 
ACTUAL FLOW RATE TO CONDENSER 1  [ft^3/hr]

*  VWCO2: 
ACTUAL FLOW RATE TO CONDENSER 2  [ft^3/hr]

*  PPC: 
PUMP POWER FOR ON CONDENSER PUMP, [kW]

*  PMC: 
TOTAL POWER CONSUMPTION FOR MAIN CONDENSER PUMPS  [kW]

*  TOPA: 
TOWER OPERATION MODE

*           -3: 
AUTOMATIC CONTROL OF TOWER OPERATION

*           -2: 
ONLY TOWER B OPERATED

*           -1: 
ONLY TOWER A OPERATED

*            0: 
NO COOLING TOWER OPERATION

*            1: 
ONE COOLING TOWER PUMP AND TWO TOWERS ON

*            2: 
TWO COOLING TOWER PUMPS AND TWO TOWERS ON

*  VWA2: 
WATER FLOW RATE FOR TWO PUMP OPERATION, TOWER A  [ft^3/hr]

*  VWA1: 
WATER FLOW RATE FOR ONE PUMP OPERATION, TOWER A  [ft^3/hr]

*  VTOWA: 
ACTUAL WATER FLOW RATE, TOWER A  [ft^3/hr]

*  MTOWA: 
MASS FLOW RATE PER CELL TO TOWER A  [kg/hr]

*  VWB2: 
WATER FLOW RATE FOR TWO PUMP OPERATION, TOWER B  [ft^3/hr]

*  VWB1: 
WATER FLOW RATE FOR ONE PUMP OPERATION, TOWER B  [ft^3/hr]

*  VTOWB: 
ACTUAL WATER FLOW RATE, TOWER B  [ft^3/hr]

*  MTOWB: 
MASS FLOW RATE PER CELL TO TOWER B  [kg/hr]

*  VTOW: 
TOTAL OF ACTUAL TOWER FLOW  [ft^3/hr]

*  RTA: 
RATIO OF WATER FLOW AT ONE PUMP TO TWO PUMPS, TOWER A

*  RTB: 
RATIO OF WATER FLOW AT ONE PUMP TO TWO PUMPS, TOWER B

*  NPU: 
ACTUAL NUMBER OF OPERATING TOWER PUMPS 

*  NTCA: 
NUMBER OF TOWER CELLS, TOWER A

*  NTCB: 
NUMBER OF TOWER CELLS, TOWER B

*  PFANA: 
FAN POWER FOR ONE CELL, TOWER A   [kW]

*  PTFA: 
TOTAL FAN POWER TOWER A  [kW]

*  PFANB: 
FAN POWER FOR ONE CELL, TOWER B   [kW]

*  PTFB: 
TOTAL FAN POWER TOWER B  [kW]

*  PPT: 
POWER FOR ONE  TOWER PUMP    [kW]

*  PTPU: 
TOTAL PUMP POWER FOR TOWER OPERATION  [kW]

*  PPMU: 
POWER FOR ONE MAKE UP PUMP   [kW]

*  PMU: 
TOTAL PUMP POWER FOR MAKE UP FLOW  [kW]

*  VPMU: 
VOLUMETRIC FLOW RATE FOR ONE MAKE UP PUMP  [ft^3/hr]

*  VMU: 
TOTAL MAKE UP FLOW  [ft^3/hr]

*  PPLH: 
POWER FOR ONE LOW HEAD CIRCUIT PUMP [kW]

*  PLH: 
TOTAL PUMP POWER FOR LOW HEAD FLOW  [kW]

*  VPLH: 
VOLUMETRIC FLOW RATE FOR ONE LOW HEAD PUMP  [ft^3/hr]

*  VLH: 
TOTAL LOW HEAD FLOW RATE  [ft^3/hr]

*  VWTOT: 
TOTAL CIRCULATING COOLING WATER FLOW  [ft^3/hr]

*  VWLAKE: 
VOLUMETRIC FLOW RATE INTO LAKE  [ft^3/hr]

*  DTLH: 
DIFFERENCE IN TEMPERATURE RISE BETWEEN CONDENSER AND 

*           
LOW HEAD [F]

*  COPU1: 
NUMBER OF COOLING WATER PUMPS CONDENSER 1

*  COPU2: 
NUMBER OF COOLING WATER PUMPS CONDENSER 2

*  NPMU: 
NUMBER OF MAKE UP PUMPS

*  NPLH: 
NUMBER OF LOW HEAD PUMPS

*  TOCTR: 
CONTROL VARIABLE RETURNED FROM CONTROLLER, 0 OR 1

*  TTOA: 
COLD WATER TEMPERATURE OF TOWER A [F]

*  TTOB: 
COLD WATER TEMPERATURE OF TOWER B [F]

*  TTOW: 
MIXED TOWER OUTLET TEMPERATURE  [F]

*  TLAKE: 
LAKE RETURN TEMPERATURE  [F]

*  TCOLD: 
MIXED PLANT RETURN TEMPERATURE, TOWERS AND LAKE [F]

*  TCOUT1: 
HOT CONDENSER TEMPERATURE, CONDENSER 1  [[F]

*  TCOUT2: 
HOT CONDENSER TEMPERATURE, CONDENSER 2  [[F]

*  THOTCO: 
MIXED HOT TEMPERATURE FROM CONDENSERS 1+2  [F]

*  THOT: 
MIXED HOT PLANT DISCHARGE TEMPERATURE  [F]

*  AUXD1: 
AUXILIARY ENERGY CONSUMTION, DUMMY 1  [kW]

*  AUXD2: 
AUXILIARY ENERGY CONSUMTION, DUMMY 2  [kW]

*  PAUX: 
TOTAL AUXILIARY ENERGY CONSUMPTION  [MW]

*  PAUX1: 
AUXILIARY POWER CONSUMPTION FOR UNIT 1  [MW]

*  PAUX2: 
AUXILIARY POWER CONSUMPTION FOR UNIT 2  [MW] 

***************************************************************************

      INTEGER INFO,ICNTRL       !TRNSYS specific                                    

      DOUBLE PRECISION XIN,OUT  !TRNSYS specific

      REAL TIME,T,DTDT,PAR      !TRNSYS specific

      DIMENSION XIN(13),OUT(11),PAR(17),INFO(15) !TRNSYS specific

      DOUBLE PRECISION VWCO1,VWCO2,MTOWA,MTOWB,VTOWA,VTOWB,PTFA,PTFB,

     .      PTPU,PMU,VMU,VLH,VWTOT,VWLAKE,TOCTR,TTOA,TTOB,TTOW,

     .      TLAKE,TCOLD,TCOUT1,TCOUT2,THOTCO,THOT,AUXD1,AUXD2,PAUX,

     .      DTLH,COPU1,COPU2,NPMU,NPLH,DENS,RAC1,RAC2,TOPA,

     .      VWCD1,VWCD2,VWCP1,VWCP2,VWA1,VWA2,VWB1,VWB2,RTA,RTB,NPU,

     .      NTCA,NTCB,PFANA,PFANB,PPT,PPMU,VPMU,VPLH      

C  PARAMETERS      

C  DESIGN CONDENSER FLOW RATE AT TWO PUMP OPERATION 1+2

      VWCD1=PAR(1)

      RAC1=PAR(2)

C  RATIO OF ONE PUMP TO TWO PUMPS 1+2

      VWCD2=PAR(3)

      RAC2=PAR(4)

C  TOWER OPERATION MODE 

C  0,1,2: TOWER PUMPS, -2: AUTOM. CONTROL

C  -1: ONLY ONE TOWER

      TOPA=PAR(5)

C  TOWER A FLOW RATE AT 100% AND RATIO FOR 1 PUMP

      VWA2=PAR(6)

      RTA=PAR(7)

C  TOWER B FLOW RATE AT 100% AND RATIO FOR 1 PUMP

      VWB2=PAR(8)

      RTB=PAR(9)

C  NUMBER OF CELLS  A   B  

      NTCA=PAR(10)

      NTCB=PAR(11)

C  FAN POWER PER FAN  A+B

      PFANA=PAR(12)

      PFANB=PAR(13)

C  PUMP POWER FOR ONE TOWER PUMP

      PPT=PAR(14)

C  PUMP POWER FOR ONE MAKE UP WATER PUMP

      PPMU=PAR(15)                                    

C  VOLUMETRIC FLOW RATE PER MAKE UP PUMP

      VPMU=PAR(16)

C  VOLUMETRIC FLOW RATE PER LOW HEAD PUMP

      VPLH=PAR(17)

C  INPUTS

C  DIFFERENCE IN LOW HEAD TO MAIN CIRCUIT TEMP. RISE

      DTLH=XIN(1)

C  NUMBER OF PUMPS COND. 1+2

      COPU1=XIN(2)

      COPU2=XIN(3) 

C  NUMBER MAKE UP PUMPS

      NPMU=XIN(4)

C  NUMBER OF LOW HEAD PUMPS

      NPLH=XIN(5)

C  CONTROL VARIABLE, 0 OR 1

      TOCTR=XIN(6)

C  TOWER OUTLET TEMPERATURES  A+B

      TTOA=XIN(7)

      TTOB=XIN(8)

C  LAKE TEMPERATURE

      TLAKE=XIN(9)

C  CONDENSER OUTLET TEMPERATURE 1+2

      TCOUT1=XIN(10)

      TCOUT2=XIN(11)

C  AUXILIARY POWER DUMMY 1 AND 2

      AUXD1=XIN(12)

      AUXD2=XIN(13)

C  CONDENSER 

C  FLOW RATE AT ONE CIRCULATING PUMP

      VWCP1=RAC1*VWCD1

      VWCP2=RAC2*VWCD2

C  FLOW RATE FOR CONDENSERS, BOTH ARE TREATED AS IF THEY WERE EQUAL

      VWCO1=AMOD(COPU1,2)*VWCP1+DIM(COPU1,1)*VWCD1                

      VWCO2=AMOD(COPU2,2)*VWCP2+DIM(COPU2,1)*VWCD2                

C  LOW HEAD FLOW RATE

      VLH=NPLH*VPLH 

C  TOTAL CIRCULATING WATER FLOW

      VWTOT=VWCO1+VWCO2+VLH

C  MIXED HOT WATER TEMPERATURE FROM CONDENSERS

      THOTCO=(VWCO1*TCOUT1+VWCO2*TCOUT2)/(VWCO1+VWCO2)      

C  TOTAL HOT PLANT DISCHARGE TEMPERATURE (MIXED)

      THOT=(VWCO1*TCOUT1+VWCO2*TCOUT2+VLH*(THOTCO-DTLH))/VWTOT            

C  TOWER FLOW RATES AT 1 PUMP OPERATION

      VWA1=VWA2*RTA

      VWB1=VWB2*RTB

10    CONTINUE

C  FLOW RATE FOR TOWERS AND MIXED TOWER OUTLET TEMPERATURE

      IF (TOPA .EQ. -3.) THEN

         VTOWA=VWA1+TOCTR*(VWA2-VWA1)

         VTOWB=VWB1+TOCTR*(VWB2-VWB1)

         NPU=TOCTR+1.

      ELSE IF (TOPA .EQ. -2.) THEN

         VTOWA=0.

         VTOWB=VWB2

         NPU=1.

      ELSE IF (TOPA .EQ. -1.) THEN

         VTOWA=VWA2 

         VTOWB=0.

         NPU=1.

      ELSE IF (TOPA .EQ. 0.)  THEN

         VTOWA=0.

         VTOWB=0.

         NPU=0.  

      ELSE

         VTOWA=VWA1+(TOPA-1.)*(VWA2-VWA1)    

         VTOWB=VWB1+(TOPA-1.)*(VWB2-VWB1)

         NPU=TOPA   

      END IF 

C  TOTAL TOWER FLOW RATE

      VTOW=VTOWA+VTOWB  

C  MIXED COLD TOWER TEMPERATURE

      IF (VTOW .LE. 0.) THEN

         TTOW=TTOA

      ELSE          

         TTOW=(TTOA*VTOWA+TTOB*VTOWB)/VTOW

      END IF

C  WATER FLOW RATE INTO LAKE

      VWLAKE=VWTOT-VTOW  

C  OVERALL MIXED COLD TEMPERATURE

      TCOLD=(VWLAKE*TLAKE+VTOW*TTOW)/VWTOT      

*      IF (VTOW. LE. 0.) THEN

*         TCOLD=TLAKE

*      END IF

*      WRITE(*,'(4A12)') 'TLAKE','TCOLD','VWLAKE','VWTOT'

*      WRITE(*,'(2F12.3,2F12.1)') TLAKE,TCOLD,VWLAKE,VWTOT

C  PARAMETER CHECK

      IF (VWLAKE .LT. (0.05*VWTOT) ) THEN

        IF (TOPA .EQ. 0.) GOTO 20

        IF (TOPA .EQ. -2.)  TOPA=0.           

        IF (TOPA .EQ. -1.)  TOPA=0.

        IF (TOPA .EQ.  1.)  TOPA=0.

        IF (TOPA .EQ. -3.)  TOPA=1.

        IF (TOPA .EQ.  2.)  TOPA=1.

C        WRITE(*,*) 'NEW PARAMETER:',TOPA

        GOTO 10 

20      CONTINUE

      END IF  

C  CALCULATION OF AUXILIARY POWER

C  TOWER POWER CONSUMPTION

C     FANS 

      PTFA=NTCA*PFANA

      PTFB=NTCB*PFANB

      IF (TOPA .EQ. -2.) PTFA=0.

      IF (TOPA .EQ. -1.) PTFB=0.

      IF (TOPA .EQ. 0.) THEN

         PTFA=0.

         PTFB=0.

      END IF

C     PUMPS

      PTPU=NPU*PPT

C  MAKE UP ENERGY CONSUMPTION

      PMU=NPMU*PPMU

C  TOTAL AUXILIARY POWER CONSUMPTION

      PAUX=PTFA+PTFB+PTPU+PMU+AUXD1+AUXD2

C  DEFINE OUTPUT VARIABLES TO OTHER COMPONENTS

C  COOLING TOWER SPECIFIC

C  MASS FLOW RATES PER TOWER CELL

C  WATER DENSITY 

      DENS=62.2 !lb/ft^3       

      IF (VTOWA .EQ. 0.) VTOWA=50000*0.1337*60 !ft^3/hr

      MTOWA=VTOWA*DENS*0.4536/NTCA  !kg/hr

      IF (VTOWB .EQ. 0.) VTOWB=50000*0.1337*60 !ft^3/hr

      MTOWB=VTOWB*DENS*0.4536/NTCB  !kg/hr

C  TOTAL AUXILIARY POWER

      PAUX=PAUX/1000  !kW => MW

C  TOTAL MAKE UP FLOW 

      VMU=NPMU*VPMU      

C  MASS FLOW RATES TOWER A + B

      OUT(1)=MTOWA  !kg/hr

      OUT(2)=MTOWB  !kg/hr    

C  VOLUMETRIC FLOW RATES

C  LAKE      

      OUT(3)=VWLAKE !ft^3/hr

C  MAKE UP FLOW

      OUT(4)=VMU    !ft^3/hr      

C  CONDENSERS      

      OUT(5)=VWCO1  !ft^3/hr

      OUT(6)=VWCO2  !ft^3/hr

C  MIXED HOT RETURN TEMPERATURE FROM CONDENSERS AND LOW HEAD      

      OUT(7)=THOT   !F

C  MIXED COLD TEMPERATURE FROM TOWERS       

      OUT(8)=TTOW   !F               

C  MIXED COLD TEMPERATURE FROM LAKE AND TOWERS      

      OUT(9)=TCOLD  !F

C  AUXILIARY POWER CONSUMPTION

      OUT(10)=PAUX  !MW

C  NUMBER OF TOWER PUMPS

      OUT(11)=NPU   

      RETURN 1

      END   

D 3 Type 75 Condenser

D 3.1 Description of Type 75 condenser model
The condenser model type 75 is used in the calculation to calculate the condenser shell pressure and cooling water outlet temperature from the cooling water inlet temperature and standard load (see chapter 5). The model is adapted to the Columbia Generating Station and contains some parameters that were obtained from measurements at the power plant. The parameters that were used in the calculation are:

Cleanliness factor
AC=0.55  BC=0 1/MW

Steam flow rate
AL=-206.15 klb/hr  BL=7.058 klb/hr_MW

Subcooling

AS=15.21 ˚F  BS=-0.01466 ˚F/MW

Following are the parameters, inputs and outputs of the type 75 routine:

Parameters:

1
Intercept for steam flow rate


[klb/hr]

2
Linear coefficient for steam flow rate

[klb/hr-MW]

3
Intercept for sub cooling



[˚F]

4
Linear coefficient for sub cooling

[˚F/MW]

5
Intercept for cleanliness factor

6
Linear coefficient for cleanliness factor

7
Design water flow rate



[ft3/hr]

Inputs:

1
Cooling water inlet temperature


[˚F]

2
Corresponding load at standard condition
[MW]

3
Volumetric water flow rate


[ft3/hr]

Outputs:

1
Outlet temperature



[˚F]

2
Condenser shell temperature


[˚F]

3
Back pressure




["Hg]

      SUBROUTINE TYPE75(TIME,XIN,OUT,T,DTDT,PAR,INFO,ICNTRL,*) 

************************************************************************

*  THIS TYPE SIMULATES THE CONDENSER OF A STEAM CYCLE POWER PLANT.

*  INPUTS ARE THE COOLING WATER INLET TEMPERATURE [F], THE EQUIVALENT

*  LOAD AT STANDARD CONDITIONS [MW] AND THE CONDENSER WATER FLOW RATE 

*  [ft^3/hr]. 

*  OUTPUTS ARE:

*  1 CONDENSER BACK PRESSURE ["Hg]

*  2 CONDENSER SHELL TEMPERATURE [F] 

*  3 COOLING WATER OUTLET TEMPERATURE [F].

*  PARAMETERS ARE CURVE FIT COEFFICIENTS FOR SUBCOOLING, STEAM FLOWRATE

*  AND CLEANLINESS FACTOR AND THE DESIGN WATER FLOW RATE. 

****************************************************************************         

****************************************************************************

* NOMENCLATURE

* A:      
OVERALL CONDENSER SURFACE AREA [ft^2] 

* A_P:   
CROSSECTIONAL AREA OF FLOW PATH, SUM OF ALL CONDENSER PIPES [ft^2]

* AC:     
INTERCEPT FOR CLEANLINESS FACTOR RELATION

* AL:     
INTERCEPT FOR THROTTLE STEAM FLOW RATE RELATION [klb]

* AS:     
INTERCEPT FOR SUBCOOLING RELATION [F]

* BC:     
LINEAR COEFFICIENT FOR CLEANLINESS FACTOR RELATION [1/MW] 

* BL:     
LINEAR COEFFICIENT FOR THROTTLE STEAM FLOW RATE RELATION [klb/MW]

* BS:     
LINEAR COEFFICIENT FOR SUBCOOLING RELATION [F/MW]

* C:      
EMPIRICAL TUBE SIZE CONSTANT

* C_F:   
CLEANLINESS FACTOR OF CONDENSER TUBES

* C_M:  
MATeRIAL CONSTANT

* C_T:   
WATER INLET TEMPERATURE CORRECTION

* C8-C13: COEFFICIENTS FOR SATURATION PRESSURE CALCULATION

* CW:   
SPECIFIC HEAT OF WATER [BTU/lb*F]

* DENS: 
AVERAGE DENSITY OF WATER [lb/ft^3]

* DTSUB: SUBCOOLING OF CONDENSATE [F]

* HEX:  
EXIT ENTHALPY OF STEAM AFTER HIGH PRESSURE TURBINE [BTU/lb] 

* LOAD_ST:EQUIVALENT POWER PLANT UNIT LOAD AT STANDARD CONDITION [MW]

* MST:  
EXHAUST STEAM FLOW INTO CONDENSER [lb/hr]

* MSTT:
THROTTLE STEAM FLOW RATE AFTER BOILER BEFORE 

*         
HIGH PRESSURE TURBINE [lb/hr]

* MW:   
MASS FLOW RATE OF COOLING WATER [lb/hr]

* PHG:  
CONDENSER BACK PRESSURE [inHg] 

* PSAT: 
CONDENSER BACK PRESSURE [PSIA]

* R:   
RATIO OF THROTTLE FLOW TO CONDENSER FLOW [lb/lb]

* RAT: 
RATIO OF OVERALL HEAT TRANSFER COEFFICIENT TO WATER CAPACITANCE 

*

RATE

* TIN:    
COOLING WATER INLET TEMPERATURE [F]

* TOU:    COOLING WATER OUTLET TEMPERATURE [F]

* TR:     
CONDENSER SHELL TEMPERATURE [R]

* TS:     
CONDENSER SHELL TEMPERATURE [F]

* U:      
HEAT TRANSFER COEFFICIENT [BTU/hr*F*ft^2]

* UA:     
OVERALL HEAT TRANSFER COEFFICIENT [BTU/hr*F]

* V_P:  
WATER VELOCITY IN CONDENSER TUBES [ft/s]

* V_W:  
VOLUMETRIC FLOW RATE OF COOLING WATER [ft^3/hr]

* V_WDES: VOLUMETRIC FLOW RATE OF COOLING WATER AT 100% FLOW [ft^3/hr]

* X:      
NUMERATOR OF SHELL TEMPERATURE RELATION

* Y:      
DENOMINATOR OF SHELL TEMPERATURE RELATION 

* XIN:    
INPUT ARRAY

* OUT:  
OUTPUT ARRAY

* PAR:   
PARAMETER ARRAY

* TIME,T,DTDT: TRNSYS SPECIFIC, NOT USED IN THIS CALCULATION 

****************************************************************************

      INTEGER INFO,ICNTRL       !TRNSYS specific                                    

      DOUBLE PRECISION XIN,OUT  !TRNSYS specific

      REAL TIME,T,DTDT,PAR      !TRNSYS specific 

      DOUBLE PRECISION A, A_P, AC, AL, AS, BC, BL, BS, C, C_F, C_M, C_T, 

     .                 CW, DENS, DTSUB, HEX, LOAD_ST, MST, MSTT,  

     .                 MW,PHG, PSAT, R, RAT, TIN, TOUT,TR,TS,U,UA,V_P, 

     .                 V_W, V_WDES, X, Y

      DIMENSION XIN(3),OUT(4),PAR(7),INFO(15) !TRNSYS specific

C  DECLARATION OF CONSTANTS FOR SATURATION PRESSURE CALCULATION

      DATA C8/-1.04404E+4/,C9/-1.129464E+1/

      DATA C10/-2.70224E-2/,C11/1.28904E-5/,C12/-2.47807E-9/

      DATA C13/6.54597/                                        

C  INPUT VARIABLES

      TIN=XIN(1)           !F,       WATER INLET

      LOAD_ST=XIN(2)       !MW,      EQUIVALENT PLANT LOAD AT STANDARD CONDITION

      V_W=XIN(3)           !ft^3/hr, WATER FLOW RATE

C  PARAMETERS

      AL=PAR(1)            !klb/hr

      BL=PAR(2)            !klb/hr-MW

      AS=PAR(3)            !F

      BS=PAR(4)            !F/MW

      AC=PAR(5)

      BC=PAR(6)            !1/MW

      V_WDES=PAR(7)        !ft^3/hr

C  CALCULATE HEAT TRANSFER COEFFICIENT

C  cleanliness factor from polynimial fit

      C_F=AC+BC*LOAD_ST

C  empirical pipe constant

      C=263 !for 1" pipes

C  crossectional area of all pipes in condenser

      A_P=59.43051 !ft^2

C  water velocity in pipes

      V_P=V_WDES/3600/A_P !ft/s

C  material correction factor

      C_M=1.0

C  inlet water temperature correction 

      C_T=8.2956E-03+2.1398e-02*TIN-1.0199e-04*TIN**2

C  heat transfer coefficient

      U=C*SQRT(V_P)*C_F*C_M*C_T      

C  condenser surface area

      A=250000 !ft^2

C  overall heat transfer coefficient

      UA=U*A

C  STEAM THROTTLE FLOW RATE AS A FUNCTION OF LOAD

      MSTT=(AL+BL*LOAD_ST)*1000  !lb/hr 

C  STEAM MASS FLOW RATE INTO CONDENSER 

C  ratio of throttle steam flow to condenser flow

C      R=0.468+1.7014E-3*TIN

      R=0.66

C  steam into condenser

      MST=R*MSTT   !lb/hr

C  ASSUME CONSTANT EXHAUST ENTHALPY

      HEX=1038.3 !BTU/lb

C  SUBCOOLING OF CONDENSATE AS FUNCTION OF LOAD

      DTSUB=AS+BS*LOAD_ST

C  WATER MASS FLOW RATE

      DENS=62.2    !lb/ft^3

      MW=V_W*DENS  !lb/hr

C  SPECIFIC HEAT OF WATER, ASSUMED CONSTANT

      CW=0.997

C  RATIO OF WATER HEATING TO LOG MEAN TEMPERTURE DIFFERENCE

      RAT=UA/(MW*CW)

C  CALCULATE WATER OUTLET TEMPERATURE

      X=(HEX+CW*DTSUB-(CW*TIN/(1-EXP(RAT)))+(MW/MST*CW*TIN)+CW*32)

      Y=MW/MST*CW-EXP(RAT)/(1-EXP(RAT))*CW

      TOUT=X/Y

C  CONDENSER SHELL TEMPERATURE

      TS=(TIN-TOUT*EXP(RAT))/(1-EXP(RAT))

C  PRESSURE IN CONDENSER 

      TR=TS+459.67 !F to R

      PSAT=EXP(C8/TR+C9+C10*TR+C11*TR**2+C12*TR**3+C13*ALOG(TR)) !psia

      PHG=PSAT*2.036  

C  OUTPUT VARIABLES

      OUT(1)=TOUT   !F

      OUT(2)=TS     !F

      OUT(3)=PHG    !"Hg

      RETURN 1

      END 

